Q:

Erro em fórmula com OU

Alguma razão para isto não funcionar?
Eu pretendo que ele conte quantas situações existem dentro de um ou outro critério.
Tanto esta fórmula:
=CONTAR.SE.S(E:E;">120";E:E;"<139" OU F:F;">80";F:F;"<89")

Como esta:
=CONTAR.SE.S((E:E;">120";E:E;"<139")OU(F:F;">80";F:F;"<89"))

Não funciona.
Traduzindo melhor o que pretendo é assim:
Eu quero o valor total de uma tabela em que os valor da coluna E se encontram entre 120 e 139 ou entao que na coluna F os valores se encontrem entre 80 e 89.
Caso essa linha tenha as duas situações como verdadeiras apenas vai contar uma.
Tipo um if, se condicao 1=verdadeiro entao conta 1 passa para o proximo caso contrario verifica a condicao 2 e se esta for verdadeira conta 1 e passa para o proximo

A:

Atenção: Não sei se isso se deve à versão do Excel, mas no meu Excel
  em Português (versão 14.0.7149.5000 32 bits do Office 2010) não
  existe uma função chamada CONTAR.SE.S. A função que existe no meu
  Excel é CONT.SES (e por isso eu a uso no exemplo). Ainda assim,
  supondo que a sua função esteja com o nome correto, a resposta
  continua válida (basta trocar o nome da função).

A função CONT.SES, de contagem com múltiplos critérios, permite que você faça a contagem usando múltiplis critérios e múltiplos intervalos. Então, pra ter o cálculo que você deseja bastan fazer três contagens:

Conte os valores na coluna E que estejam dentro dos intervalos de interesse dessa coluna
Conte os valores na coluna F que estejam dentro dos intervalos de interesse dessa coluna
Conte os valores em ambas colunas que estejam dentro dos seus respectivos intervalos.

Se você somar os valores de (1) e (2), ele vai estar quase sempre certo, exceto quando ocorrerem as exceções que você menciona (os valores de ambas as colunas E e F em uma determinada linha serem ambos aceitos). Essas exceções são contabilizadas por (3), então basta subtrair esse valor da soma anterior.
Por exemplo:
=CONT.SES(E:E;">120";E:E;"<139")+CONT.SES(F:F;">80";F:F;"<89")-CONT.SES(E:E;">120";E:E;"<139";F:F;">80";F:F;"<89")

Pra evitar que os valores fiquem fixos na fórmula, talvez você deseje configurar outras células com esses valores (por exemplo, as células I1, J1, K1 e L1). Nesse caso, faça a fórmula assim:
=CONT.SES(E:E;">"&I1;E:E;"<"&J1)+CONT.SES(F:F;">"&K1;F:F;"<"&L1)-CONT.SES(E:E;">"&I1;E:E;"<"&J1;F:F;">"&K1;F:F;"<"&L1)

Eis um exemplo de planilha (que você poderia ter fornecido - facilitiaria enormemente a vida de quem fosse lhe ajudar), com o resultado (as células "Em E", "Em F" e "Em E e F" contém as fórmulas parciais para demonstrar os resultados em cada item anteriormente exposto). A linha indicada pela seta em vermelho é a única em que as condições são satisfeitas nas duas colunas.

